

Penggunaan MAPLE untuk Pembelajaran ALJABAR

Drs. Sahid, MSc.

Lab Komputer Jurdik Matematika FMIPA UNY

Pendahuluan

Maple merupakan paket aplikasi matematika yang dapat digunakan untuk melakukan berbagai perhitungan matematis baik secara eksak (analitik) maupun numerik. Dengan kemampuan yang dimiliki, Maple merupakan sebuah alat bantu yang handal untuk pemecahan masalah matematika, baik masalah komputasi numerik, aljabar simbolik, maupun visualisasi (grafik). Sesungguhnya, Maple tidak hanya berguna untuk melakukan perhitungan matematis saja, namun juga dapat digunakan sebagai editor teks untuk menghasilkan dokumen yang memuat penjelasan atau uraian verbal dan berbagai perhitungan matematis. Dengan kemampuan visualisasi matematis interaktif, sebuah antarmuka grafis tempat menuliskan masukan dan menampilkan keluaran yang menyerupai notasi matematika yang sesungguhnya, fasilitas pengolahan kata, dan bahasa pemrograman, Maple telah digunakan oleh jutaan pemakai di seluruh dunia di kalangan pendidikan, lembaga riset, dan industri.

Makalah ini menguraikan bagaimana pemakaian Maple untuk menyelesaikan masalah-masalah aljabar. Diharapkan dengan uraian ini, peserta pelatihan (para guru) dapat menggunakan Maple untuk membantu pembelajaran Matematika di SMU, khususnya tentang topik Aljabar. Pembahasan setiap topik dalam makalah ini memuat **contoh-contoh** yang dikerjakan secara lengkap untuk memperkenalkan pemakaian setiap perintah Maple, dan **latihan-latihan** yang dapat Anda kerjakan langsung pada layar Maple. Pada bagian akhir tutorial juga diberikan **Soal-Soal Latihan**, untuk memberikan kesempatan bagia Anda mencoba menggunakan perintah-perintah Maple yang sudah Anda pelajari.

Sebelum memasuki bagian yang membahas bagaimana mengerjakan matematika dengan Maple, berikut disajikan uraian tentang lingkungan tempat Anda bekerja dengan Maple.

Mengenal Lingkungan Kerja Maple

Berinteraksi dengan Mesin Komputasi Maple

Maple menggunakan dua buah komponen, yakni **Kelompok Eksekusi** (*Execution groups*) dan **Tabel** (*spreadsheets*), yang membantu pemakainya berinteraksi dengan mesin komputasi Maple. Kedua komponen tersebut merupakan sarana utama bagi pemakai untuk memerintahkan Maple melakukan perintah dan menampilkan hasilnya. Perintah-perintah Maple dapat dituliskan pada kedua komponen tersebut.

Kelompok Eksekusi (Execution Groups)

Kelompok eksekusi merupakan unsur komputasi dasar di dalam lembar kerja Maple. Elemen tersebut merupakan gabungan satu atau lebih perintah Maple beserta hasil (output)-nya sebagai satu kesatuan yang dapat dijalankan ulang dengan sekali menekan tombol ENTER pada

saat kursor berada di suatu kelompok eksekusi. Sebuah kelompok eksekusi di dalam lembar kerja Maple ditandai dengan sebuah tanda kurung siku di sebelah kiri baris perintah. Apabila tanda kurung kelompok eksekusi tidak ada, dapat ditampilkan dengan menggunakan menu **View --> Show Group ranges**. Setiap kelompok eksekusi pada Maple ditandai dengan > (**tanda lebih besar**). Berikut adalah sebuah contoh kelompok eksekusi yang berisi sebuah perintah (ekspresi) Maple dan hasilnya.

> **solve(a*x^2=4,{x});**

$$\left\{x = 2 \frac{1}{\sqrt{a}}\right\}, \left\{x = -2 \frac{1}{\sqrt{a}}\right\}$$

Pada Maple, Anda dapat menuliskan perintah **solve** tersebut. Apabila kursor berada di baris dalam sebuah kelompok eksekusi dan Anda menekan tombol ENTER, maka semua perintah dalam kelompok eksekusi akan dikerjakan oleh Maple dan ditampilkan hasilnya di bawahnya, seperti contoh di atas. Kursor secara otomatis akan berpindah ke kelompok eksekusi berikutnya.

Selain perintah-perintah (biasanya ditulis warna merah) Maple dan hasilnya (biasanya berwarna biru), sebuah kelompok eksekusi dapat memuat paragraf yang berisi penjelasan tambahan (teks biasa atau notasi matematika yang bukan perintah Maple).

Contoh

Kelompok eksekusi di bawah ini memuat sebuah paragraf dan sebuah baris perintah Maple yang berisi satu perintah (ekspresi) Maple. Seperti di atas, pada Maple ketika kursor berada di baris perintah dan Anda menekan ENTER, hasilnya akan ditampilkan.

Sebuah perintah (ekspresi) Maple dan hasil perhitungannya.


> **expand((a+b)^3);**

$$a^3 + 3 a^2 b + 3 a b^2 + b^3$$

Pada saat Anda baru saja menjalankan Maple atau membuka dokumen baru dengan perintah **File --> New**, lembar kerja Maple hanya berisi sebuah kelompok eksekusi yang memuat sebuah baris perintah (seperti ditunjukkan di bawah ini). Untuk membuat kelompok eksekusi baru di dalam lembar kerja Maple, Anda dapat menggunakan menu **Insert --> Execution Group --> [Before Cursor | After Cursor]**.

Selain perintah (ekspresi) Maple, sebuah baris perintah dapat memuat sebuah **komentar** untuk penjelasan singkat. Sebuah komentar yang ditulis di baris perintah **harus** diawali dengan tanda **#**. Setiap ekspresi yang ditulis di belakang tanda **#** tidak akan dikerjakan oleh Maple. Perhatikan contoh di bawah ini. Sebuah perintah Maple dapat menghasilkan output yang berupa ekspresi numerik, simbolik, atau grafik. Contoh berikut ini memberikan tampilan hasil yang berupa permukaan tiga dimensi.

> **plot3d(sin(x*y),x=-2..2,y=-1..1); # permukaan z=sin(xy) setelah diputar**



Terdapat dua hal yang perlu Anda perhatikan di dalam menuliskan perintah-perintah (ekspresi) Maple:

1. Setiap baris perintah (tepatnya, setiap ekspresi Maple) harus diakhiri dengan tanda **titik koma (;)** agar hasilnya dapat ditampilkan. Maple akan memberikan pesan ERROR apabila suatu ekspresi tidak diakhiri dengan tanda titik koma.
2. Apabila Anda tidak ingin segera menampilkan hasil sebuah perintah Maple, akhiri perintah tersebut dengan tanda **titik dua (:)**. Cara ini berguna untuk menampilkan hasil (output) Maple di lain tempat. Sudah tentu Anda harus menyimpannya ke dalam sebuah variabel agar hasil tersebut dapat Anda panggil di tempat lain.

Tabel Komputasi (*Spreadsheets*)

Maple memungkinkan penggunaanya untuk menampilkan tabel (lembar kerja seperti MS Excel) yang memuat ekspresi-ekspresi numerik maupun simbolik. Fasilitas ini memberikan kemudahan untuk menampilkan tabel rumus, misalnya. Berikut adalah contoh sebuah tabel yang berisi kolom ekspresi matematis, notasi integral, integral tak tentu, dan turunan (derivatif). Anda dapat merujuk sebuah sel seperti pada MS Excel, yakni dengan menuliskan nomor kolom (A, B, C, ...) dan nomor baris (1, 2, 3, ...) diawali tanda **tilde (~)**, misalnya **~A3** untuk merujuk sel A3, di dalam setiap ekspresi Maple.

	A	B	C	D
1	$f(x)$	$\int f(x) e^x dx$	<i>Hasil Integral</i>	$\frac{\partial}{\partial x} f(x) e^x$
2	1	$\int e^x dx$	e^x	e^x
3	x	$\int x e^x dx$	$x e^x - e^x$	$e^x + x e^x$
4	$\sin(x)$	$\int \sin(x) e^x dx$	$-\frac{1}{2} \cos(x) e^x + \frac{1}{2} \sin(x) e^x$	$\cos(x) e^x + \sin(x) e^x$
5	$\cos(x)$	$\int \cos(x) e^x dx$	$\frac{1}{2} \cos(x) e^x + \frac{1}{2} \sin(x) e^x$	$-\sin(x) e^x + \cos(x) e^x$
6	x^2	$\int x^2 e^x dx$	$x^2 e^x - 2 x e^x + 2 e^x$	$2 x e^x + x^2 e^x$
7	$\ln(x)$	$\int \ln(x) e^x dx$	$\ln(x) e^x + \text{Ei}(1, -x)$	$\frac{e^x}{x} + \ln(x) e^x$

Menuliskan dan Mengeksekusi Perintah Maple

Bagian ini menyajikan tiga alat utama untuk menuliskan perintah Maple dan menampilkan hasilnya.

Menuliskan Perintah-perintah Maple

Seperti sudah Anda ketahui, perintah (ekspresi) Maple dituliskan pada baris perintah (>) diakhiri dengan tanda titik koma (;). Untuk menjalankan suatu ekspresi, tempatkan kursor pada baris tersebut dan tekan tombol ENTER.

Perintah Maple dapat ditampilkan dalam bentuk notasi Maple (memuat perintah-perintah Maple, seperti **exp(x)**) atau dalam bentuk notasi matematika baku (misalnya, e^x).

Menggunakan Menu Peka-Konteks

Dengan menggunakan hasil atau ekspresi Maple yang sudah ada, Anda dapat melakukan tindakan baru. Untuk melihat daftar tindakan yang dapat Anda lakukan terhadap sebuah objek Maple, **klik kanan** objek Maple tersebut. Anda akan melihat sebuah menu peka-konteks, karena isinya tergantung objek yang sedang Anda tunjuk.

Sekarang boleh Anda coba lakukan. Jalankan Maple Anda, tuliskan perintah berikut ini di layar Maple.


> **z=sin(x^2*y);**

$$z = \sin(x^2 y)$$

1. Arahkan mouse Anda ke baris perintah (yang berwarna merah) kemudian **klik kanan**, menu apa saja yang Anda lihat? Pilih salah satu, misalnya menu **Execute**. Anda akan melihat tampilan outputnya dalam bentuk ekspresi matematika baku. Anda dapat mencoba menu yang lain jika mau!
2. Sekarang arahkan mouse Anda ke ekspresi output (yang berwarna biru), kemudian **klik kanan**, menu apa saja yang ditampilkan? Samakah dengan menu sebelumnya? Tidak bukan? Pilih salah satu menu, misalnya **Plots --> 3D Implicit Plot --> x,y,z**. Anda akan melihat sebuah baris perintah baru dan outputnya berupa gambar yang merupakan permukaan dengan persamaan tersebut, seperti di bawah ini. (Anda akan mendapatkan hasil dari menu peka-konteks langsung di bawah objek yang Anda tunjuk!)

> **smartplot3d[x,y,z](z=sin(x^2*y));**

Live



Anda dapat mencoba menu-menu peka-konteks yang lain untuk mengetahui hasilnya dengan objek yang sama.

Geret dan Taruh (*Dragging and Dropping*)


Anda dapat memindahkan output Maple dari atau ke Maple plot. Jika Anda memindah hasil sebuah persamaan ke sebuah Maple plot, persamaan tersebut akan ditampilkan sebagai plot kurva atau permukaan. Sebaliknya, jika Anda memindah sebuah kurva atau permukaan dari Maple plot ke tempat lain, maka persamaan kurva atau permukaan tersebut akan ditampilkan. Cobalah Anda lakukan dengan memindah (dengan cara mendrag bukan copy & paste) gambar di atas ke tempat lain. Selanjutnya, coba menggunakan perintah di bawah ini, pindahkan outputnya ke gambar di atas, Anda lihat hasilnya.

> **z=sin(x^2)*y^2;**

$$z = \sin(x^2) y^2$$

> `smartplot3d[x,y,z](z = sin(x^2)*y^2);`

Live



Pengaturan Dokumen

Seperti sudah disebutkan, Maple bukanlah sekedar alat bantu untuk perhitungan matematika, namun juga berguna untuk menghasilkan naskah yang syarat dengan perhitungan-perhitungan matematika. Sebuah dokumen Maple, selain memuat ekspresi Maple dan outputnya (kelompok eksekusi), juga dapat memuat paragraf dan **hyperlink**. Berikut dijelaskan bagaimana cara pengaturan dokumen di dalam Maple.

Paragraf dan Teks

Sebuah paragraf di dalam dokumen Maple analog dengan paragraf pengolah kata biasa. Paragraf memuat teks, notasi matematika, grafik, termasuk duplikat output Maple, seperti plot. Sebuah paragraf juga dapat ditulis di dalam kelompok eksekusi, seperti sudah dijelaskan di atas.

Paragraf berisi uraian atau penjelasan, seperti yang sedang Anda baca ini. Anda dapat mengatur format paragraf dan tulisan, seperti halnya di pengolah kata biasan. Berikut adalah contoh-contoh pengaturan paragraf dan teks serta tulisan yang dapat Anda lakukan di dalam dokumen Maple.

1. Paragraf terpusat.
2. Paragraf rata kiri.
3. Paragraf rata kanan.
4. Teks dapat dicetak dengan *huruf miring*, digaris bawah, **cetak tebal**, atau ***kombinasi***.
5. *Anda juga dapat mengatur ukuran dan jenis huruf sesuka Anda!*

Seperti halnya pada pengolah kata dan **deskstop publisher**, Anda dapat membuat format paragraf dan teks Anda sendiri, dengan menggunakan menu **Format --> Styles**.

Anda dapat menuliskan notasi matematika dan menampilkan grafik di dalam sebuah paragraf, sehingga Anda dapat menulis artikel tentang matematika secara lebih komprehensif. Untuk menuliskan notasi matematika di dalam paragraf, tekan tombol **Ctrl+R**, Maple akan berganti ke modus notasi matematika. Untuk berganti ke modus teks tekan **Ctrl+T**. Simbol-simbol matematika dapat dihasilkan dengan perintah-perintah yang mirip dengan perintah-perintah LaTeX, namun tanpa diawali dengan garis miring ke kanan. Sebagai contoh, Anda dapat menuliskan paragraf sebagai berikut dengan mudah:

Untuk menghitung integral tentu $\int_0^{\beta} x^2 \sin(\alpha x) dx$ dengan Maple, tuliskan perintah:

```
> int(x^2*sin(alpha*x),x=0..beta );
```

Bandingkan dengan cara menuliskan notasi matematika di MS Word, misalnya, mana yang lebih mudah?

Kalkulasi Numerik

Maple dapat melakukan perhitungan matematis secara eksak maupun numerik. Maple dapat digunakan sebagai kalkulator, bahkan dapat melakukan semua bentuk perhitungan dalam matematika. Pada bab ini Anda akan belajar menggunakan Maple sebagai alat kalkulasi secara numerik. Tulis perintah-perintah Maple di bawah ini (yang dicontohkan di belakang prompt [`>`] kemudian tekan tombol ENTER).

Perhitungan Aritmetika Eksak

Pada level yang paling dasar, Maple dapat digunakan sebagai kalkulator yang sangat handal. Untuk melakukan perhitungan matematis, tulis ekspresi matematika pada baris perintah Maple (di belakang tanda [`>`], diakhiri dengan tanda titik koma (;). Setelah Anda menekan tombol ENTER, maka Maple akan menampilkan hasilnya. Untuk membuat ekspresi matematis di Maple Anda dapat menggunakan operasi-operasi : + (**penjumlahan**), - (**pengurangan**), * (**perkalian**), / (**pembagian**), ^ (**pangkat**). Setiap ekspresi Maple yang diakhiri dengan tanda titik koma (;) akan ditampilkan hasilnya. Apabila suatu ekspresi diakhiri dengan tanda titik dua (:), maka hasil perhitungannya tidak ditampilkan oleh Maple.

Contoh 1:

Tulis contoh-contoh di bawah ini pada baris prompt Maple dan ikuti dengan menekan tombol ENTER. Anda akan melihat hasil yang ditampilkan Maple sama seperti yang terlihat di bawah ini.

```
> 2+4;
```

6

```
> 12*34567890;
```

414814680

Contoh 2:

Menghitung 134^{39} :

```
> 134^39;
```

```
905914344031473705525163856620677712914023509111870374238564740740974  
23209059057664
```

Untuk menghitung $(32)(12^{13})$ Anda dapat menuliskan ekspresi berikut ini:

```
> 32*12^13;
```

3423782572130304

Maple memberikan jawaban eksak sebanyak 83 digit!

Contoh 3:

Maple dapat menghitung pecahan tanpa mengubah ke pecahan desimal:

```
3/5 + 5/9 + 7/12;
```

$$\frac{313}{180}$$

Contoh 4:

Menghitung akar kuadrat suatu bilangan dengan fungsi **sqrt()** :

> **sqrt(24);**

$$2\sqrt{6}$$

Maple menyederhanakan ke bentuk yang paling sederhana.

Contoh 5:

Maple menyediakan konstanta-konstanta penting dalam matematika. Untuk menggunakan nilai π ketik **Pi**.

> **4*(3+Pi);**

$$12 + 4\pi$$

Lagi, Maple menggunakan nilai eksak untuk π , bukan hampiran numeriknya.

Contoh 6:

Maple juga memberikan nilai eksak dalam perhitungan fungsi trigonometri.

> **sin(5*Pi/3);**

$$-\frac{1}{2}\sqrt{3}$$

> **sec(Pi/4);**

$$\sqrt{2}$$

> **arcsin(-1);**

$$-\frac{1}{2}\pi$$

Jika Anda menulis nama fungsi yang tidak dikenal oleh Maple, akan muncul pesan *error*:

> **tan(Pi/2);**

Error, (in tan) numeric exception: division by zero

Contoh 7:

Menghitung fungsi eksponen e^x .

> **exp(x);**

$$e^x$$

> **exp(1);**

$$e$$

Contoh 8:

Menghitung harga mutlak $|x|$.

> **abs(x);**

$$|x|$$

> **abs(-3);**

$$3$$

> **abs(exp(1)-Pi);**

$$-e + \pi$$

Contoh 9:

Faktorisasi prima sebuah bilangan asli dengan fungsi **ifactor()**:

> **ifactor(31722722304);**

$$(2)^{10} (3) (7)^2 (13)^2 (29) (43)$$

Contoh 10:

Anda dapat menuliskan beberapa ekspresi matematika dalam satu baris, setiap ekspresi matematis diakhiri dengan tanda titik koma. Semua ekspresi akan dikerjakan setelah Anda menekan tombol ENTER.

> **sin(Pi/3); cos(Pi/3); tan(Pi/3);**

$$\frac{1}{2}\sqrt{3}$$

$$\frac{1}{2}$$

$$\sqrt{3}$$

Contoh 11:

Anda dapat menghasilkan suatu barisan bilangan dengan perintah **seq(..)**.

Berikut adalah contoh untuk menghasilkan barisan 20 kuadrat bilangan asli pertama.

> **seq(k^2,k=1..20);**

1, 4, 9, 16, 25, 36, 49, 64, 81, 100, 121, 144, 169, 196, 225, 256, 289, 324, 361, 400

Contoh 12:

Maple mengenal berbagai operator khusus, termasuk operator faktorial, FPB (faktor persekutuan terbesar), KPK (kelipatan persekutuan terkecil), perhitungan modulus, dan lain-lain. Berikut adalah beberapa contoh yang dapat Anda coba.

> **20!;**

2432902008176640000

Anda dapat menggunakan hasil ekspansi 20! tersebut di dalam perhitungan berikutnya tanpa harus menuliskannya. Untuk menggunakan nilai pada **hasil perhitungan terakhir**, gunakan tanda persen (%). Sebagai contoh, perintah **ifactor** dengan argumen tanda persen di bawah ini akan memfaktorkan 200! ke dalam bentuk faktorisasi prima.

> **ifactor(%);**

$$(2)^{18} (3)^8 (5)^4 (7)^2 (11) (13) (17) (19)$$

Perintah di bawah ini menghasilkan hasil kali faktor-faktor tersebut, yakni ekspansi 200! semula.

> **expand(%);**

2432902008176640000

Contoh 13: Menghitung banyaknya digit (angka satuan) dalam suatu (lambang) bilangan desimal.

> **length(32145);**

5

> **length(10^7);**

8

Dapatkah Anda menghitung cacah digit dalam bilangan hasil ekspansi 200! (dua ratus faktorial)?

> **length(200!);**

Aritmetika Bilangan Riil (Titik Mengambang)

Kekuatan utama Maple adalah kemampuannya melakukan perhitungan matematis secara eksak. Selama proses perhitungan, pecahan dan akar tidak diubah ke hampiran (nilai) desimal, sehingga terhindar dari galat pembulatan. Jika Anda memerlukan nilai-nilai desimal, Maple memiliki sebuah perintah untuk menampilkan hampiran nilai suatu ekspresi ke dalam bentuk bilangan riil (titik mengambang atau *floating-point*).

Hampiran Numerik dengan menggunakan perintah `evalf()`

Maple juga dapat digunakan untuk melakukan perhitungan hampiran secara numerik dengan menggunakan perintah `evalf()`. Dengan menggunakan perintah `evalf`, Anda akan mendapatkan hampiran nilai tersebut dalam bentuk bilangan riil (titik mengambang atau *floating-point*).

Contoh 1:

Bandingkan hasil kedua perintah di bawah ini.

> `3/5+5/9+7/12;`

$$\frac{313}{180}$$

> `evalf(3/5+5/9+7/12);`

1.738888889

Contoh 2:

Untuk menyimpan hasil perhitungan ke dalam sebuah variabel, digunakan tanda titik dua dan sama dengan (`:=`), seperti contoh di bawah ini.

> `k:=3/5+5/9+7/12;`

$$k := \frac{313}{180}$$

> `evalf(k);`

1.738888889

Catatan penting: Maple membedakan huruf besar dan huruf kecil, k tidak sama dengan K .

> `k;`

$$\frac{313}{180}$$

> `K;`

K

> `joe:=2^5;`

$joe := 32$

> `sqr(joe);`

$$4\sqrt{2}$$

Contoh 3:

Untuk menampilkan nilai numerik dengan cacah digit tertentu, tuliskan banyaknya digit pada perintah `evalf()` seperti contoh di bawah ini.

> `w:=4*(3+Pi);`

$w := 12 + 4\pi$

> evalf(w);

24.56637062

> evalf(w,4);

24.57

> evalf(w,45);

24.5663706143591729538505735331180115367886776

Contoh 4:

Pemakaian titik desimal pada ekspresi matematis akan menyebabkan Maple memberikan hasil numerik secara langsung. Bandingkan contoh-contoh di bawah ini.

> sqrt(34);

$\sqrt{34}$

> sqrt(34.0);

5.830951895

> 4-1/3;

$\frac{11}{3}$

> 4.0-1/3;

3.666666667

Contoh 5:

Menggunakan perintah **evalf()** pada serangkaian hasil perhitungan.

> barisan:=seq(sqrt(k),k=1..10);

$barisan := 1, \sqrt{2}, \sqrt{3}, 2, \sqrt{5}, \sqrt{6}, \sqrt{7}, 2\sqrt{2}, 3, \sqrt{10}$

> evalf(barisan);

1., 1.414213562, 1.732050808, 2., 2.236067978, 2.449489743, 2.645751311,
2.828427124, 3., 3.162277660

Menampilkan Hasil Perhitungan Sebelumnya dengan tanda persen (%):

Perhatikan ekspresi $\frac{2^{30}\sqrt{3}}{3^{20}}$, yang dapat ditulis pada baris perintah Maple sebagai berikut.

> 3/5+5/9+7/12;

$\frac{313}{180}$

> (2^30/3^20)*sqrt(3);

$\frac{1073741824}{3486784401}\sqrt{3}$

> evalf(%);

.5333783739

> evalf(%%);

1.738888889

> Pi;

π

> evalf(%);

3.141592654

> %5;

8.141592654

Jumlah dan Hasil Kali Berhingga dan Tak Berhingga

Maple dalam digunakan untuk menghitung jumlag deret dengan berhingga maupun tak berhingga suku.

Perhatikan jumlah deret berhingga suku $\sum_{i=1}^{10} \frac{1+i}{1+i^4}$, yang dapat dituliskan ke dalam baris perintah Maple sebagai berikut.

> **restart:**

> **Sum((1+i)/(1+i^4),i=1..10);**

$$\sum_{i=1}^{10} \frac{1+i}{1+i^4}$$

Untuk mengetahui nilainya gunakan perintah **value**.

> **value(%);**

$$\frac{51508056727594732913722}{40626648938819200088497}$$

Perhatikan deret tak hingga $\sum_{k=1}^{\infty} \frac{1}{k^2}$, yang dapat dituliskan pada baris perintah Maple sebagai berikut.

> **Sum(1/k^2, k=1..infinity);**

$$\sum_{k=1}^{\infty} \frac{1}{k^2}$$

> **%=value(%);**

$$\sum_{k=1}^{\infty} \frac{1}{k^2} = \frac{1}{6} \pi^2$$

Maple juga dapat menghitung perkalian dengan cacah faktor berhingga maupun tak berhingga.

Untuk menghitung hasil kali berhingga faktor $\prod_{i=0}^{10} \frac{i^2+3i-11}{i+3}$, Anda dapat menuliskan perintah-perintah Maple sebagai berikut.

> **Product(((i^2+3*i-11)/(i+3)), i=0..10);**

$$\prod_{i=0}^{10} \frac{i^2+3i-11}{i+3}$$

> **%=value(%);**

$$\prod_{i=0}^{10} \frac{i^2 + 3i - 11}{i + 3} = \frac{-7781706512657}{40435200}$$

Anda dapat melakukan perhitungan bilangan riil dengan menggunakan tingkat keakuratan yang Anda inginkan. Maple dapat melakukan perhitungan sampai ratusan ribu digit pada kebanyakan sistem operasi. Perintah berikut ini menampilkan hasil perhitungan sebelumnya sampai lima puluh digit.

> `%%=evalf(%%,50);`

$$\prod_{i=0}^{10} \frac{i^2 + 3i - 11}{i + 3} = -192448.81965854008388730610952833175055397277619500$$

Bilangan Kompleks dan Fungsi-fungsi Khusus

Maple juga dapat digunakan untuk melakukan berbagai perhitungan bilangan kompleks. Maple menyimpan bilangan imajiner $\sqrt{-1}$ dengan simbol **I** (huruf I besar).

> `(3+5*I)/(7+4*I);`

$$\frac{41}{65} + \frac{23}{65}I$$

Anda dapat dengan mudah mengubah bilangan kompleks ke bentuk koordinat kutub dengan menggunakan perintah

convert. Maple menyajikan hasilnya ke dalam bentuk **polar**(r, θ) dengan r adalah jari-jari modulus dan θ sudut kutubnya.

> `convert(%,polar);`

$$\text{polar}\left(\frac{1}{65} \sqrt{2210}, \arctan\left(\frac{23}{41}\right)\right)$$

Anda dapat melakukan perhitungan secara numerik nilai-nilai fungsi elementer dan berbagai fungsi khusus serta konstanta khusus dengan Maple. Sebagai contoh untuk menghitung nilai konstanta e (basis logaritma alam) sampai 40 digit Anda dapat menuliskan perintah sebagai berikut.

> `evalf(exp(1.0),40);`

$$2.718281828459045235360287471352662497757$$

Perhatikan fungsi Gamma, untuk menghitung nilai $\Gamma(2.5)$ tulis perintah berikut ini.

> `evalf(GAMMA(2.5));`

$$1.329340388$$

Akhirnya, jika Anda ingin menampilkan nilai π sampai 100 digit tuliskan perintah sebagai berikut.

> `evalf(Pi,100);`

$$3.141592653589793238462643383279502884197169399375105820974944592307816406286208998628034825342117068$$

Menghapus Nilai suatu Variabel

Untuk mengganti nilai sebuah variabel, gunakan variabel yang sama untuk menyimpan ekspresi matematis lain.

> `h;`

$$h$$

> `h:=56;`

$h := 56$

> **h;**

56

> **h:=sqrt(Pi);**

$h := \sqrt{\pi}$

> **h;**

$\sqrt{\pi}$

> **x:=65;**

$x := 65$

Jika kita menuliskan ekspresi seperti di bawah ini, Maple secara otomatis memasukkan nilai x ke dalam ekspresi tersebut.

> **w:=x^2-4*x+7;**

$w := 3972$

Untuk mendefinisikan suatu variabel umum, tulis ekspresi seperti di bawah ini.

> **x:='x';**

$x := x$

> **w:=x^2-4*x+7;**

$w := x^2 - 4x + 7$

Perintah restart berguna untuk menghapus semua variabel yang ada di dalam memori Maple.

> **p:=4;**

$p := 4$

> **p; x; h;**

4

x

$\sqrt{\pi}$

Sekarang lakukan dan perhatikan hasilnya.

> **restart;**

> **p; x; h;**

p

x

h

Soal-soal Latihan

- **Latihan 1** Hitunglah 37^{43} dengan Maple.
- **Latihan 2** Hitunglah $\sqrt{34}$ sampai 18 digit.
- **Latihan 3** Hitunglah hampiran numerik : $\frac{3 + \pi}{7 - \sqrt{13}}$
- **Latihan 4** Tuliskan ekspresi-ekspresi di bawah ini dan perhatikan hasilnya setelah Anda menekan tombol ENTER.

```
> 4+Pi;
> evalf(%);
> %+10;
> %+10;
```

Dapatkan Anda menjelaskan mengapa kedua ekspresi terakhir memberikan hasil yang berbeda?

```
> a:=4+Pi;
> b:=evalf(a);
> b+10;
```

Kalkulasi Aljabarik

Maple merupakan software "C.A.S" (**Computer Algebra System**). Artinya, Maple tahu setiap aturan aljabar dalam matematika.

```
> restart;
```

Mengekspansi, Memfaktorkan, dan Menyederhanakan Ekspresi

Maple menyediakan berbagai cara yang berbeda untuk memanipulasi dan menampilkan ekspresi-ekspresi matematis, sehingga memudahkan Anda melihat, menjelaskan, dan menggunakannya secara efektif. Fleksibilitas ini memungkinkan Anda melakukan hal-hal seperti mengekspansi suku-suku binomial, memfaktorkan hasilnya, menyederhanakan ekspresi-ekspresi trigonometrik, menyimpan hasilnya ke dalam suatu variabel, dan mengkonversi suatu ekspresi ke bentuk lain, dan sebagainya.

Mengekspansi dan Memfaktorkan Suatu Ekspresi

Contoh 1:

Gunakan perintah **expand()** untuk mengekspansi $(x+2)^2(3x-3)(x+5)$.

```
> k:=(x+2)^2*(3*x-3)*(x+5);
```

$$k := (x + 2)^2 (3x - 3) (x + 5)$$

```
> expand(k);
```

$$3x^4 + 24x^3 + 45x^2 - 12x - 60$$

Contoh 2:

Perintah **expand()** juga dapat digunakan untuk menguraikan fungsi-fungsi $\sin(2x)$ dan $\cos(2x)$.

```
> expand(sin(2*x));
```

$$2 \sin(x) \cos(x)$$

```
> expand(cos(2*x));
```

$$2 \cos(x)^2 - 1$$

Cobalah ekspansikan ekspresi-ekspresi: $\sin(3x)$, $\cos(6x)$, dsb.

Contoh 3:

Ekspansikan ekspresi $x^{\left(\frac{1}{2}\right)} \left(x^{\left(\frac{3}{2}\right)} + x^{\left(-\frac{1}{2}\right)} \right)$.

```
> h:=x^(1/2)*(x^(3/2)+x^(-1/2));
```

$$h := \sqrt{x} \left(x^{(3/2)} + \frac{1}{\sqrt{x}} \right)$$

```
> expand(h);
```

$$x^2 + 1$$

Contoh 4:

Maple dapat mengekspansi perpangkatan binomial seperti $(x + y)^{15}$. Perintah-perintah Maple di bawah ini akan menyimpan ekspresi tersebut dan mengekspansinya.

> **binom15:=(x+y)^15;**

$$\text{binom15} := (x + y)^{15}$$

> **expand(binom15);**

$$\begin{aligned} &x^{15} + 15 y x^{14} + 105 y^2 x^{13} + 455 y^3 x^{12} + 1365 y^4 x^{11} + 3003 y^5 x^{10} + 5005 y^6 x^9 + 6435 y^7 x^8 \\ &+ 6435 y^8 x^7 + 5005 y^9 x^6 + 3003 y^{10} x^5 + 1365 y^{11} x^4 + 455 y^{12} x^3 + 105 y^{13} x^2 \\ &+ 15 y^{14} x + y^{15} \end{aligned}$$

Contoh 5:

Setelah melihat hasilnya, Anda dapat menggunakan perintah **factor** untuk memfaktorkannya dan menguji kebenaran hasilnya.

> **factor(%);**

$$(x + y)^{15}$$

Contoh 6:

Faktorkan ekspresi $3x^2 - 10x - 8$

> **w:=3*x^2-10*x-8;**

$$w := 3x^2 - 10x - 8$$

> **factor(w);**

$$(3x + 2)(x - 4)$$

Secara langsung:

> **factor(3*x^2-10*x-8);**

$$(3x + 2)(x - 4)$$

Contoh 7:

Ekspansikan ekspresi $2(x - 2)(2x^2 + 5x + 2)(x + 4)$. Gunakan perintah **factor()** untuk memfaktorkan

kembali. Bandingkan hasilnya, mengapa tampak berbeda?

> **H:=2*(x-2)*(2*x^2+5*x+2)*(x+4);**

$$H := 2(x - 2)(2x^2 + 5x + 2)(x + 4)$$

> **ans:=expand(H);**

$$\text{ans} := 4x^4 + 18x^3 - 8x^2 - 72x - 32$$

> **factor(ans);**

$$2(x - 2)(2x + 1)(x + 4)(x + 2)$$

Contoh 8:

Maple dapat memfaktorkan ekspresi dengan lebih dari satu variabel. Faktorkan ekspresi:

$$x^2y + 2xy + y$$

> **h:=x^2*y+2*x*y+y;**

$$h := x^2y + 2xy + y$$

> **factor(h);**

$$y(x+1)^2$$

Contoh 9:

Jika Maple tidak dapat memfaktorkan sebuah ekspresi, hasilnya sama dengan ekspresi semula.

> **factor(3*x^2-10*x-9);**

$$3x^2 - 10x - 9$$

Contoh 10:

Maple juga dapat memfaktorkan fungsi-fungsi non-polinomial, misalnya $\sin^2 x - \cos^2 x$.

> **factor((sin(x))^2-(cos(x))^2);**

$$-(\cos(x) - \sin(x))(\cos(x) + \sin(x))$$

Contoh 11:

Maple juga dapat memfaktorkan ekspresi rasional.

> **A:=(x^3-7*x^2+15*x-9)/(x^2+4*x+4);**

$$A := \frac{x^3 - 7x^2 + 15x - 9}{x^2 + 4x + 4}$$

> **factor(A);**

$$\frac{(x-1)(x-3)^2}{(x+2)^2}$$

> **B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);**

$$B := \frac{x^3 - 7x^2 + 15x - 9}{x^2 - 4x + 3}$$

> **factor(B);**

$$x - 3$$

Contoh 12:

Perintah **numer()** berguna untuk mendapatkan *pembilang* dalam sebuah ekspresi pecahan, dan fungsi **denom()** mendapatkan *penyebut* sebuah pecahan.

> **B:=(x^3-7*x^2+15*x-9)/(x^2-4*x+3);**

$$B := \frac{x^3 - 7x^2 + 15x - 9}{x^2 - 4x + 3}$$

> **factor(numer(B)); factor(denom(B));**

$$(x-1)(x-3)^2$$

$$(x-1)(x-3)$$

Menyederhanakan Ekspresi Matematis

Contoh 1:

Maple dapat menyederhanakan ekspresi rumit seperti,

$$\cos(x)^5 + \sin(x)^4 + 2 \cos(x)^2 - 2 \sin(x)^2 - \cos(2x).$$

> **V:=cos(x)^5+sin(x)^4+2*cos(x)^2-2*sin(x)^2-cos(2*x);**

$$V := \cos(x)^5 + \sin(x)^4 + 2 \cos(x)^2 - 2 \sin(x)^2 - \cos(2x)$$

> **simplify(V);**

$$\cos(x)^5 + \cos(x)^4$$

Contoh 2:

Sederhanakan $\sin(x)^5 + \cos(x)^4$.

> **simplify(sin(5*t)+sin(3*t));**

$$16 \sin(t) \cos(t)^4 - 8 \sin(t) \cos(t)^2$$

Contoh 3:

Menyederhanakan ekspresi $\frac{1}{x+1} + \frac{x}{x-1}$.

> **M:=(1/(x+1))+x/(x-1));**

$$M := \frac{1}{x+1} + \frac{x}{x-1}$$

> **simplify(M);**

$$\frac{2x - 1 + x^2}{x^2 - 1}$$

Cara lain untuk menyederhanakan ekspresi pecahan adalah dengan perintah **normal**, yang akan menghilangkan faktor persekutuan pada pembilang dan penyebut.

Contoh 4:

Pecahan $\frac{x^3 - y^3}{x^2 + x - y - y^2}$ dapat disederhanakan dengan perintah Maple sebagai berikut.

> **normal((x^3-y^3)/(x^2+x-y-y^2));**

$$\frac{y^2 + xy + x^2}{y + x + 1}$$

Menyimpan Nilai ke dalam Variabel

Seperti sudah Anda lihat contoh-contoh sebelumnya, Anda dapat menyimpan suatu ekspresi atau hasil perhitungan ke dalam sebuah variabel dengan menggunakan tanda titik dua dan sama dengan (**:=**). Pemakaian variabel sangat membantu jika Anda bekerja dengan banyak ekspresi dan fungsi, serta jika Anda menggunakan ulang hasil-hasil perhitungan sebelumnya.

Contoh 1:

Simpan ekspresi $3x^2 + 8$ ke dalam variabel **W**.

> **W:=3*x^2+8;**

$$W := 3x^2 + 8$$

Untuk menghitung nilai $3x^2 + 8$ untuk $x=4$, gunakan perintah **subs()**:

> **subs(x=4,3*x^2+8);**

$$56$$

Cara lain adalah menggunakan **subs()** pada **W**.

> **subs(x=4,W);**

$$56$$

Contoh 2:

Perintah **subs()** juga dapat digunakan untuk substitusi variabel. Untuk mengganti x pada ekspresi $3x^2 + 8$ dengan $5 + 2u$, tulis ekspresi di bawah ini. Hasil substitusinya disimpan ke dalam variabel **M**.

> **W:=3*x^2+8;**

$$W := 3x^2 + 8$$

> **M:=subs(x=5+2*u,W);**

$$M := 3 (5 + 2 u)^2 + 8$$

Maple dapat mengekspansi sebuah ekspresi dengan perintah **expand()**.

> **expand(M);**

$$83 + 60 u + 12 u^2$$

Contoh 3:

Perintah **subs()** juga dapat digunakan untuk substitusi pada ekspresi dengan satu variabel atau lebih.

Ekspresi berikut mengganti x dengan 7 dan y dengan 12 pada ekspresi $U = \frac{2}{5}x^2 + 3y$.

> **U:=(2/5)*x^2+3*y;**

$$U := \frac{2}{5}x^2 + 3y$$

> **subs(x=7,y=12,U);**

$$\frac{278}{5}$$

> **evalf(%);**

$$55.60000000$$

Contoh 4:

Perintah **subs()** dapat digunakan untuk memasukkan sebuah nilai ke dalam persamaan, untuk menguji apakah nilai tersebut memenuhi persamaan tersebut. Misalnya diketahui persamaan $x^3 - 5x^2 + 7x - 12 = 0$. Ekspresi-ekspresi berikut berguna untuk menguji apakah nilai x yang dimasukkan memenuhi persamaan tersebut.

> **pers:=x^3-5*x^2+7*x-12=0;**

$$eqn := x^3 - 5x^2 + 7x - 12 = 0$$

> **subs(x=3,pers);**

$$-9 = 0$$

> **subs(x=4,pers);**

$$0 = 0$$

> **subs(x=5,pers);**

$$23 = 0$$

Misalkan variabel **P1** menyatakan ekspresi suku banyak $(41x^2 + x + 1)^2(2x - 1)$.

> **P1:=(41*x^2+x+1)^2*(2*x-1);**

$$P1 := (41x^2 + x + 1)^2(2x - 1)$$

Untuk menguraikan ekspresi tersebut dapat digunakan perintah **expand** dengan argumen (masukan) **P1**. Misalkan hasilnya disimpan ke dalam variabel **P2**.

> **P2:=expand(P1);**

$$P2 := 3362x^5 - 1517x^4 + 84x^3 - 79x^2 - 1$$

Anda dapat menghitung nilai **P1** dan **P2** untuk $x = 1$ dengan menggunakan perintah **eval**. Samakah hasilnya?

> **eval(P1,x=1); eval(P2,x=1);**

1849

1849

Jadi, selain menggunakan perintah **subs**, Anda juga dapat menggunakan perintah **eval** untuk menghitung nilai suatu ekspresi yang memuat variabel.

Pada contoh berikut kita gunakan variabel **Q** untuk menyimpan hasil bagi suatu polinomial dengan polinomial dalam bentuk **normal** (pecahan paling sederhana).

> **pembilang:=P2;**

$$\text{pembilang} := 3362x^5 - 1517x^4 + 84x^3 - 79x^2 - 1$$

> **penyebut:=expand((3*x+5)*(2*x-1));**

$$\text{penyebut} := 6x^2 + 7x - 5$$

> **Q:=normal(pembilang/penyebut);**

$$Q := \frac{1681x^4 + 82x^3 + 83x^2 + 2x + 1}{3x + 5}$$

Fungsi: Definisi, Nilai dan Grafik

Untuk membedakan fungsi dengan ekspresi, Maple memerlukan notasi khusus untuk mendefinisikan sebuah fungsi. Maple menyediakan beberapa cara untuk mendefinisikan fungsi. Salah satu cara adalah menggunakan notasi anak panah (\rightarrow), yang menyerupai notasi fungsi atau pemetaan dalam matematika. Perintah **unapply** dapat digunakan untuk mengubah suatu ekspresi menjadi fungsi.

Misalnya untuk mendefinisikan fungsi $f(x) = \cos(\pi x) + 3$ pada Maple digunakan perintah:

> **f:=x->cos(Pi*x)+3;**

$$f := x \rightarrow \cos(\pi x) + 3$$

Perhatikan cara menuliskannya dengan menggunakan notasi " \rightarrow " untuk mendefinisikan fungsi $x \rightarrow f(x)$. Maple tidak akan mendefinisikan fungsi jika Anda menuliskan $f(x):=\cos(\text{Pi}*x)+3$; Bandingkan ekspresi dan fungsi di bawah ini.

> **y:=(x+2)/(x^3+5*x+2);**

$$y := \frac{x+2}{x^3+5x+2}$$

> **f:=x->(x+2)/(x^3+5*x+2);**

$$f := x \rightarrow \frac{x+2}{x^3+5x+2}$$

Mendefinisikan fungsi $x \rightarrow x^2 + \frac{1}{2}$.

> **f := x -> x^2+1/2 ;**

$$f := x \rightarrow x^2 + \frac{1}{2}$$

Untuk menghitung nilai fungsi, cara yang digunakan persis sama dengan cara dalam matematika. Maple akan menghasilkan nilai simbolik (eksak).

> **f(2);**

> **f(a+b);**

$$\frac{9}{2}$$

$$(a+b)^2 + \frac{1}{2}$$

Perintah **unapply** dapat digunakan untuk mengubah suatu ekspresi menjadi fungsi.

> **g := unapply(x^2 + 1/2, x); g(2);**

$$g := x \rightarrow x^2 + \frac{1}{2}$$

$$\frac{9}{2}$$

Latihan 1 Definisikan fungsi $h(x) = x^3 \sin(2x + 1)$.

Jawaban 1

> **h:= x-> x^3*sin(2*x+1);**

Setelah sebuah fungsi didefinisikan, kita dapat menggunakannya untuk menghitung nilainya di titik tertentu, atau menggambar grafiknya.

> **f:=x->ln(cos(5*x));**

> **f(x);**

Membatalkan definisi fungsi:

> **f:='f';**

Menghitung Nilai Fungsi

> **f:=x->3*x+x^2;**

$$f := x \rightarrow 3x + x^2$$

> **f(-1);**

$$-2$$

> **f(2+sqrt(5));**

$$6 + 3\sqrt{5} + (2 + \sqrt{5})^2$$

> **evalf(f(2+sqrt(5)));**

$$30.65247584$$

> **f(x+4);**

$$3x + 12 + (x + 4)^2$$

> **simplify(%);**

$$11x + 28 + x^2$$

> **(f(x+h)-f(x))/h;**

$$\frac{3h + (x+h)^2 - x^2}{h}$$

> **simplify(%);**

$$3 + 2x + h$$

Anda dapat bekerja dengan lebih dari satu buah fungsi.

> **g:=x->cos(x)+1;**

$$g := x \rightarrow \cos(x) + 1$$

> **f(g(Pi/3));**

$$\frac{27}{4}$$

> **j:=x->g(f(x));**

$$j := x \rightarrow g(f(x))$$

> **j(x);**

$$\cos(3x + x^2) + 1$$

Latihan 2 Definisikan fungsi $s(t) = \frac{3+t^2}{\sqrt{3t+1}}$ kemudian hitunglah $s(2)$, $s(t-3)$, $s(t) - s(3)$ dan sederhanakan hasilnya.

Jawaban 2

> **s:=t->(3+t^2)/(sqrt(3*t+1));**

> **s(2);**

> **s(t-3);**

> **simplify(%);**

> **s(t) - s(3);**

> **simplify(%);**

Pemakaian fungsi membuat Anda tidak perlu lagi menggunakan perintah subs.

Menyelesaikan Persamaan Fungsi

> **g:='g';**

$$g := g$$

> **g:=t->t^3-6*t^2+6*t+8;**

$$g := t \rightarrow t^3 - 6t^2 + 6t + 8$$

> **solve(g(t)=0,t);**

$$4, 1 + \sqrt{3}, 1 - \sqrt{3}$$

> **fsolve(g(t)=0,t);**

$$-.7320508076, 2.732050808, 4.$$

Grafik Fungsi

Fungsi **plot()** juga dapat digunakan pada fungsi:

> **h:='h'; y:='y'; x:='x';**

$$h := h$$


$$y := y$$

$$x := x$$

> **h:=x->x*exp(-x);**

$$h := x \rightarrow x e^{(-x)}$$

> **plot(h(x),x=-1..4,y=-2..1);**




Beberapa fungsi dapat digambar sekaligus. Perhatikan fungsi $f(x) = \frac{2}{x^2 + 1}$. Berikut akan digambar grafik fungsi $f(x+1)$, $f(x-3)$ dan $f(x-6)$. Dapatkah Anda menjelaskan hubungan fungsi-fungsi tersebut?

> **f:=x->2/(x^2+1);**

$$f := x \rightarrow 2 \frac{1}{x^2 + 1}$$

> **plot([f(x),f(x+1),f(x-3),f(x-6)],x=-5..10,y=-1..3);**



Latihan 3 Definisikan fungsi $f(x) = 2x - |x^2 - 5|$ kemudian jawab pertanyaan-pertanyaan di bawah ini.

- Hitung nilai $f(6.5)$
- Sederhanakan $f(z-4)$ dengan z adalah suatu variabel.
- Plot grafik fungsi $f(x)$
- Carilah semua nilai x yang memenuhi $f(x)=0$.

Jawaban 3

> **f:=x->2*x-abs(x^2-5);**

> **f(6.5);**

> **simplify(f(z-4));**

> **plot(f(x),x);**

> **fsolve(f(x)=0,x=0..2);**

> **fsolve(f(x)=0,x=3..4);**

Latihan 4 Definisikan fungsi $g(x) = 5e^{(-.5x)}$ dan $h(x) = x + 1$ kemudian kerjakan:

- Plot grafik kedua fungsi dalam sebuah sistem koordinati. Lakukan eksperimen dengan nilai-nilai domain dan jangkauan yang berbeda.
- Estimasikan koordinat titik potong kedua kurva dengan mengklik tombol mouse kiri.
- Gunakan perintah **fsolve()** untuk menyelesaikan persamaan $g(x)=h(x)$. Bagaimanakah hubungan penyelesaian persamaan terakhir dengan jawaban Anda pada (b).

Jawaban 4

```
> g:=x->5*exp(-0.5*x);
> h:=x->x+1;
> plot([g(x),h(x)],x=-5..5,y=-20..20);
> plot([g(x),h(x)],x=1..2,y=1..4);
> x0:=fsolve(g(x)=h(x),x);
```

Penyelesaian persamaan $g(x)=h(x)$ merupakan absis titik potong kurva $y=g(x)$ dan $y=h(x)$. Untuk mencari oordniat titik potong, hitung nilai kedua fungsi pada titik absis tersebut.

```
> g(x0);
> h(x0);
```

Latihan 5 Definisikan fungsi $k(x) = x + 3 \sin(2x)$, kemudia lakukan:

- Plot grafik fungsi ini pada interval $[-1, 8]$.
- Ubah grafik pada (a) dengan memasukkan kurva $y=4$. Gunakan untuk mengestimasi nilai-nilai x yang memenuhi $k(x)=4$.
- Sebutkan fungsi yang grafiknya memberikan informasi sama dengan (b)
- Gunakah perintah **fsolve()** untuk menghitung hampiran penyelesaian $k(x)=4$.

Jawaban 5

```
a)
> k:=x->x+3*sin(2*x);
> plot(k(x),x=-1..8);
b)
> plot([k(x),4],x=-1..8);
```

Tampak adanya tiga titik potong di $x=3.25, 4.825$ dan 5.95 .

c) Kita gambar grafik $k(x) - 4$ dan lihat titik potongnya dengan sumbu- x .

```
> plot(k(x)-4,x=-1..8);
Berikut kita gunakan fsolve( ) :
> fsolve(k(x)=4,x=2 .. 3.5);
> fsolve(k(x)=4,x=3.5 .. 5);
> fsolve(k(x)=4,x=5 .. 7);
```

Menyelesaikan Persamaan dan Sistem Persamaan/Pertidaksamaan

Maple dapat digunakan untuk menyelesaikan dan menguji kebenaran penyelesaian berbagai persamaan aljabarik dan sistem persamaan.

Menyelesaikan Persamaan Aljabarik

Perintah Maple **solve()** berguna untuk menyelesaikan persamaan secara eksak (jika mungkin). Perintah Maple **fsolve()** berguna untuk menghitung penyelesaian persamaan secara numerik. Kedua perintah dapat digunakan untuk menyelesaikan sistem persamaan.

Berikut adalah salah satu cara untuk menyelesaikan persamaan

$$x^3 - \frac{ax^2}{2} + \frac{13x^2}{3} = \frac{13ax}{6} + \frac{10x}{3} - \frac{5a}{3}$$

dengan Maple:

> **pers1 := x^3-1/2*a*x^2+13/3*x^2 = 13/6*a*x+10/3*x-5/3*a;**
 > **solve(pers1, {x});**

$$\{x = \frac{2}{3}\}, \{x = -5\}, \{x = \frac{1}{2}a\}$$

Untuk menguji kebenaran penyelesaian tersebut, kita dapat menghitung kesamaan kedua ruas untuk setiap nilai x yang dihasilkan oleh Maple, misalnya

> **eval(pers1, x=1/2*a);**

$$\frac{13}{12}a^2 = \frac{13}{12}a^2$$

Rumus 'abc' untuk persamaan kuadrat:

> **solve(a*x^2+b*x+c=0,{x});**

$$\{x = \frac{1}{2} \frac{-b + \sqrt{b^2 - 4ac}}{a}\}, \{x = \frac{1}{2} \frac{-b - \sqrt{b^2 - 4ac}}{a}\}$$

> **{x = 1/2/a*(-b+sqrt(b^2-4*a*c))}, {x = 1/2/a*(-b-sqrt(b^2-4*a*c))};**

$$\{x = \frac{1}{2} \frac{-b + \sqrt{b^2 - 4ac}}{a}\}, \{x = \frac{1}{2} \frac{-b - \sqrt{b^2 - 4ac}}{a}\}$$

Memasukkan dan memanipulasi persamaan: Perintah lhs() dan rhs()

Contoh 1:

Simpan persamaan $x^3 - 5x^2 + 23 = 2x^2 + 4x - 8$ dengan nama "pers1".

> **pers1:=x^3-5*x^2+23=2*x^2+4*x-8;**

$$pers1 := x^3 - 5x^2 + 23 = 2x^2 + 4x - 8$$

Contoh 2:

Pisahkan ruas kanan dan kiri dengan perintah-perintah lhs() dan rhs().

> **lhs(pers1);**

$$x^3 - 5x^2 + 23$$

> **rhs(pers1);**

$$2x^2 + 4x - 8$$

Contoh 3 :

Perintah lhs() dan rhs() dapat digunakan untuk menyusun persamaan semula, hasilnya disimpan ke dalam variabel "pers2".

> **pers2:=lhs(pers1)-rhs(pers1)=0;**

Menghitung Penyelesaian Eksak dengan Perintah solve()

Contoh 1:

Selesaikan persamaan polinomial $3x^3 - 4x^2 - 43x + 84 = 0$.

> **solve(3*x^3-4*x^2-43*x+84=0,x);**

$$3, -4, \frac{7}{3}$$

Perhatikan, Maple memberikan tiga buah akar.

Contoh 2:

Penyelesaian suatu persamaan dapat disimpan ke dalam sebuah vektor, untuk dapat digunakan pada keperluan lain.

> **N:=solve(x^2-5*x+3=0,x);**

$$N := \frac{5}{2} + \frac{1}{2}\sqrt{13}, \frac{5}{2} - \frac{1}{2}\sqrt{13}$$

> **N[1];**

$$\frac{5}{2} + \frac{1}{2}\sqrt{13}$$

Contoh 3:

Penyelesaian persamaan yang tersimpan dalam sebuah variabel.

> **pers1:=7*x^3-11*x^2-27*x-9=0;**

$$pers1 := 7x^3 - 11x^2 - 27x - 9 = 0$$

> **H:=solve(pers1,x);**

$$H := -1, 3, \frac{-3}{7}$$

Uji bahwa setiap nilai tersebut memenuhi persamaan semula, dengan perintah **subs()**.

> **subs(x=H[1],pers1);**

$$0 = 0$$

> **subs(x=H[2],pers1);**

$$0 = 0$$

> **subs(x=H[3],pers1);**

$$0 = 0$$

Contoh 4:

Penyelesaian yang rumit dari persamaan $x^3 - 34x^2 + 4 = 0$.

> **pers1:=x^3-34*x^2+4=0;**

$$pers1 := x^3 - 34x^2 + 4 = 0$$

> **H:=solve(pers1,x);# Terlalu rumit, tidak ditampilkan di sini.**

Dalam penyelesaian di atas I menyatakan $\sqrt{-1}$. Apabila penyelesaian rumit seperti itu, lebih baik selesaikan secara numerik, dengan **evalf()**.

> **evalf(H);**

$$33.99653908 - .2 \cdot 10^{-9} I, -.3412885427 - .8590254040 \cdot 10^{-8} I, \\ .3447494447 + .8730254040 \cdot 10^{-8} I$$

Contoh 5:

Selesaikan persamaan $5 e^{\left(\frac{x}{4}\right)} = 43$.

> **solve(5*exp(x/4)=43,x);**

$$4 \ln\left(\frac{43}{5}\right)$$

Contoh 6:

Terkadang Maple tidak menampilkan semua penyelesaian sebuah persamaan. Bagaimanakah Anda dapat menuliskan semua penyelesaian persamaan $\sin(x) = \frac{1}{2}$?

> **solve(sin(x)=1/2,x);**

$$\frac{1}{6} \pi$$

Latihan 1 Selesaikan persamaan Mengapa Maple hanya menampilkan dua buah akar berlainan? Faktorkan ruas kiri untuk mengetahui akar-akarnya.

Jawaban 1

> **solve(x^3-11*x^2+7*x+147=0,x);**

> **factor(x^3-11*x^2+7*x+147);**

Penyelesaian Secara Numerik dengan perintah fsolve()

Perintah Maple **fsolve()** berguna untuk menghitung penyelesaian suatu persamaan secara numerik.

Contoh 1:

Selesaikan persamaan berikut secara numerik: $x^4 - x^3 - 17x^2 - 6x + 2 = 0$

> **pers:=x^4-x^3-17*x^2-6*x+2=0;**

$$pers := x^4 - x^3 - 17x^2 - 6x + 2 = 0$$

> **fsolve(pers,x);**

$$-3.414213562, -.5857864376, .2087121525, 4.791287847$$

Contoh 2:

Carilah semua penyelesaian riil persamaan berikut dengan perintah fsolve(): $x^3 + 1 - e^x = 0$

> **pers:=x^3+1-exp(x)=0;**

$$pers := x^3 + 1 - e^x = 0$$


> **fsolve(pers,x);**

$$0.$$

Contoh 3: Carilah penyelesaian riil lain dari persamaan $x^3 + 1 - e^x = 0$. Mula-mula gambar grafik fungsi ruas kiri.

Penyelesaian persamaan $x^3 + 1 - e^x = 0$ merupakan absis titik-titik potong grafik $y = x^3 + 1 - e^x$.

> **plot(x^3+1-exp(x),x=-3..5,y=-5..15);**



Grafik menunjukkan adanya empat titik potong. Untuk mencari penyelesaian persamaan yang terletak pada interval tertentu, tambahkan selang interval x pada perintah **fsolve()**.

> **fsolve(pers,x=-1..-2);**

-0.8251554697

> **fsolve(pers,x=1..2); fsolve(pers,x=4..5);**

1.545007279

4.567036837


Cobalah Anda menggunakan perintah **fsolve()** pada interval yang tidak memuat akar!

> **fsolve(pers,x=2..4);**

$\text{fsolve}(x^3 + 1 - e^x = 0, x, 2 .. 4)$

Maple hanya menampilkan ekspresi yang Anda tulis secara lengkap.

> **plot(x^3+1-exp(x),x=-3..50,y=-10..15);**



> **fsolve(pers,x=5..50);**

$\text{fsolve}(x^3 + 1 - e^x = 0, x, 5 .. 50)$

Tidak ada akar pada interval [5 , 50]!

> **fsolve(pers,x=-50..-1);**

$\text{fsolve}(x^3 + 1 - e^x = 0, x, -50 .. -1)$

Tidak ada akar pada interval [-50 , -1]! Jadi keempat penyelesaian persamaan adalah: -0.8251554597, 0, 1.545007279, dan 4.567036837

Contoh 4:


Selesaikan secara numerik persamaan $\frac{x^2}{20} - 10x = 15 \cos(x + 15)$

Mula-mula ubah persamaan tersebut menjadi $\frac{x^2}{20} - 10x - 15 \cos(x + 15) = 0$

> **pers:=x^2/20-10*x-15*cos(x+15)=0;**

$\text{pers} := \frac{1}{20}x^2 - 10x - 15 \cos(x + 15) = 0$

> **plot(lhs(pers),x=-10..10);**



Dari grafik terlihat adanya akar pada interval $[1, 2]$.

> **fsolve(pers,x=1..2);**

1.274092075

Latihan 2 Carilah semua penyelesaian persamaan $x^5 - 4x^3 + 3x^2 + 7x - 1 = 0$ dengan melihat grafik fungsinya terlebih dahulu.

Latihan 3 Carilah semua penyelesaian persamaan $x^2 - 2 = \ln(x + 5)$ dengan melihat grafik fungsinya terlebih dahulu. Uji penyelesaian yang diperoleh dengan perintah **subs()**.

Latihan 4 Grafik $y = 10 - x^2$ dan $y = 4 \sin(2x) + 5$ berpotongan pada interval $[-5, 5]$.

- Grafik kedua fungsi bersamaan kemudian estimasikan titik potong dengan mouse.
- Tulis sebuah persamaan yang penyelesaiannya merupakan titik-titik potong kedua kurva.
- Gunakan perintah **fsolve()** untuk menyelesaikan persamaan tersebut.
- Gunakan hasil c) untuk mengestimasi koordinat titik potong kedua kurva.
- Grafik tampak seolah ada titik potong di sekitar (1,9). Gunakan **fsolve()** dan/atau grafik pada interval di sekitar titik tersebut untuk menunjukkan bahwa tidak ada titik potong kedua kurva di sekitar lokasi tersebut.

Jawaban 4

> **y1:=10-x^2;**

> **y2:=4*sin(2*x)+5;**

a) Grafik kedua fungsi

> **plot([y1,y2],x=-5..5);**

Terdapat dua titik potong di sekitar titik-titik: (-1.8, 6.6) dan (2.75, 2) .

b) Bentuk persamaan $y_1=y_2$.

> **pers:= y1=y2;**

c) Selesaikan persamaan tersebut dengan **fsolve()**

> **x_soln1:=fsolve(y1=y2,x=-4..0);**

> **x_soln2:=fsolve(y1=y2,x=0..4);**

d) Gunakan **subs()** untuk mencari koordinat titik potong.

> **y_soln1:=subs(x=x_soln1,y1);**

> **y_soln2:=subs(x=x_soln2,y1);**

Jadi titik-titik potongnya adalah: (-1.800,6.763) dan (2.773,2.311)

e) Grafik lebih deetil di sekitar $x=1$.

> **plot([y1,y2],x=.5..1.5);**

Tidak ada titik potong di sekitar $x=1$. Pemakaian **fsolve()** juga tidak menghasilkan penyelesaian:

> **fsolve(y1=y2,x=.5..1.5);**

Menyelesaikan Persamaan Literal

> **restart:**

Selesaikan persamaan literal $4 - v = 2T - kg$ terhadap g :

> **solve(4-v=2*T-k*g,g);**

$$\frac{-4 + v + 2 T}{k}$$

Cara yang lebih baik untuk menampilkan hasilnya adalah:

> **g=solve(4-v=2*T-k*g,g);**

$$g = \frac{-4 + v + 2 T}{k}$$

Latihan 5 Selesaikan persamaan di atas terhadap variabel T , k dan v .

Penyelesaian Persamaan-persamaan Khusus

Contoh-contoh berikut ini memperlihatkan bagaimana kita dapat menggunakan Maple untuk menyelesaikan jenis-jenis persamaan lain, yakni yang melibatkan fungsi trigonometri dan harga mutlak.

Menyelesaikan persamaan trigonometri.

> **solve(arccos(x)-arctan(x)=0, {x});**

$$\left\{ x = \frac{1}{2} \sqrt{-2 + 2\sqrt{5}} \right\}$$

Menyelesaikan persamaan harga mutlak $|(z + |z + 2|)^2 - 1| = 9$.

> **solve(abs((z+abs(z+2))^2-1)^2=9,{z});**

$$\{z = 0\}, \{z \leq -2\}$$

Menyelesaikan Sistem Persamaan

Menyelesaikan Sistem Persamaan Linier (SPL) dengan **solve()**.

> **restart:**

> **with(plots):**

Warning, the name `changecoords` has been redefined

Contoh 1:

Selesaikan dua buah persamaan : $3x + 2y = 3$ dan $x - y = -4$

> **solve({3*x+2*y=3,x-y=-4});**

$$\{x = -1, y = 3\}$$

Penyelesaian tersebut merupakan titik potong kedua kurva dari masing-masing persamaan.

> **y1:=solve(3*x+2*y=3,y);**


$$y1 := -\frac{3}{2}x + \frac{3}{2}$$

> **y2:=solve(x-y=-4,y);**

$$y2 := x + 4$$

> **gb1:=plot([y1,y2],x=-5..5);**

> **gb2:=plot([[-1,3]],style=point,color=blue,symbol=circle);**
 > **display([gb1,gb2]);**



Contoh 2:

Selesaikan SPL: $\{x + y + z = 1, 3x + y = 3, x - 2y - z = 0\}$

> **solve({x+y+z=1, 3*x+y=3, x-2*y-z=0});**

$$\{x = \frac{4}{5}, y = \frac{3}{5}, z = \frac{-2}{5}\}$$

Latihan 6 Selesaikan SPL: $\{4x + 3y = 12, 5x - 7y = 35\}$. Uji hasilnya dengan memasukkannya ke dalam kedua persamaan.

Jawaban 6

> **spl:={4*x+3*y=12, 5*x-7*y=35};**

> **solusi:=solve(spl);**

> **subs(x=189/43,y=-80/43,spl);**

SPL dengan Takberhingga Penyelesaian

Contoh 1:

Selesaikan SPL : $\{x + y + z = 1, 3x + y = 3\}$.

> **solusi:=solve({x+y+z=1, 3*x+y=3});**

$$solusi := \{x = 1 + \frac{1}{2}z, y = -\frac{3}{2}z, z = z\}$$

Maple memberikan penyelesaian umum yang dinyatakan dalam x. Ganti nilai x dengan 4, misalnya.

> **subs(x=4,solusi);**

$$\{y = -\frac{3}{2}z, 4 = 1 + \frac{1}{2}z, z = z\}$$

Salah satu penyelesaian adalah : x=4, y=-9 dan z=6. Uji bahwa ini memenuhi SPL semula.

> **subs(x=2,solusi);**

$$\{y = -\frac{3}{2}z, 2 = 1 + \frac{1}{2}z, z = z\}$$

Jadi contoh dua penyelesaian adalah: $(x,y,z) = (4, -9, 6)$ dan $(2, -3, 2)$.

Latihan 7 Selesaikan SPL : $\{x + 2y + z = 2, 3x + y = 1\}$ dan tentukan sedikitnya tiga penyelesaian khusus.

Jawaban 7

> **spl:={ x+2*y+z=2 , 3*x+y=1 };**

> **solusi:=solve(spl);**

Diperoleh penyelesaian khusus untuk nilai $x = 1, 2, 3$, dan 4 . Cek kebenaran penyelesaian ini:

> **subs(x=1,spl);**

> **subs(x=2,spl);**

> **subs(x=3,spl);**

> **subs(x=4,spl);**

Menyelesaikan Sistem Pertidaksamaan

Contoh-contoh berikut memperlihatkan bagaimana mudahnya menyelesaikan sistem pertidaksamaan dengan Maple.

Berikut adalah bagaimana kita dapat menyelesaikan sistem pertidaksamaan,

$$x^2 < 1, y^2 \leq 1, x + y < \frac{1}{2}.$$

> **solve({x^2<1,y^2<=1,x+y<1/2},{x,y});**

$$\{-1 \leq y, y \leq 1, x + y < \frac{1}{2}, -1 < x, x < 1\}$$

Kita dapat menyelesaikan suatu pertidaksamaan yang melibatkan beberapa variabel, seperti $x + y + \frac{4}{x+y} < 10$. Untuk mencari penyelesaian x yang dinyatakan dalam y , dapat digunakan perintah Maple sebagai berikut.

> **ineq:=x+y+4/(x+y)<10;**

> **solve(ineq,{x});**

$$\{x < -y\}, \{5 - \sqrt{21} - y < x, x < 5 + \sqrt{21} - y\}$$

Soal-soal Latihan

Latihan 1 Simpan ekspresi $x^2 + 4x - 3$ ke dalam variabel k . Selanjutnya, simpan ekspresi $k^2 - 9$ ke dalam variabel M . Akhirnya hitung $3M + 6$.

Latihan 2 Ekspansikan $(1 + x)^4$ dengan menggunakan perintah **expand()**.

Latihan 3 Misalkan $P = ax^3 + bx^2 + cx + d$. Carilah P jika $x = .1e-1$, $a = -\frac{1}{5}$, $b = \frac{2}{5}$, $c = 0$, dan $d = \frac{13}{15}$.

Latihan 4 Gunakan perintah **subs()** untuk menguji apakah nilai-nilai $x = 1, 2$ atau 3 memenuhi persamaan $x^3 - 16x^2 + 51x - 36 = 0$.

Latihan 5 Ekspansikan $(x + 1)^n$ untuk $n = 2, 3$ dan 4 .

Latihan 6 Faktorkan ekspresi $3x^4 - 2x^3 + 22x^2 - 18x - 45$.

Latihan 7 Faktorkan ekspresi $x^{\left(\frac{1}{2}\right)} - x^{\left(\frac{3}{2}\right)}$ dan gunakan perintah **expand()** untuk menguji hasilnya.

Latihan 8 Sederhanakan $\frac{7}{x+2} + \frac{3x}{(x+2)^2}$

Latihan 9 Gunakan perintah **simplify()** pada ekspresi $\sin(3t) - \sin(7t)$, dan perhatikan hasilnya.